NAME - SANDEEP KAUR SEKHON
ROLL NO - MAT/19/112

Solving Systems of Linear Equations

A *system of linear equations* (or linear system) is a collection of one or more linear equations involving the same set of variables. A solution to a linear system is an assignment of values to the variables such that all the equations are simultaneously satisfied. The word "system" indicates that the equations are to be considered collectively, rather than individually.

A linear system may behave in any one of three possible ways:
The system has *infinitely many solutions*.
The system has a single *unique solution*.
The system has *no solution*.

Non-Homogenous systems of Linear Equations

Suppose we want to solve a system of linear equations in the form **mx=b**, where **m** is the coefficient matrix, **x** is a column vector of variables, and **b** is a column vector. Such a system is called *nonhomogeneous* when **b** is a vector with at least one nonzero entry. *Mathematica* offers several options for solving such a system, and we will explore each in turn. Let us see how *Mathematica* interprets the systems of linear equations:

$$x_1 + x_2 + x_3 = 6$$
$$2x_1 - x_2 - x_3 = -3$$
$$3x_1 - 2x_2 = 0$$

In[7]:=
```
m = {{1, 1, 1}, {2, -1, -1}, {3, -2, 0}}
```

Out[7]//MatrixForm=

$$\begin{pmatrix} 1 & 1 & 1 \\ 2 & -1 & -1 \\ 3 & -2 & 0 \end{pmatrix}$$

In[8]:=
```
x = {{x1}, {x2}, {x3}}
```

Out[8]//MatrixForm=

$$\begin{pmatrix} x1 \\ x2 \\ x3 \end{pmatrix}$$

In[9]:=
```
b = {{6}, {-3}, {0}}
```

Out[9]//MatrixForm=

$$\begin{pmatrix} 6 \\ -3 \\ 0 \end{pmatrix}$$

Using **RowReduce** and **ArrayFlatten** :

The command **ArrayFlatten** is used to form a augmented matrix, and the command **RowReduce** is used to find the reduced row echelon form of the matrix.

In[10]:=
```
ArrayFlatten[{{m, b}}]
```

Out[10]//MatrixForm=

$$\begin{pmatrix} 1 & 1 & 1 & 6 \\ 2 & -1 & -1 & -3 \\ 3 & -2 & 0 & 0 \end{pmatrix}$$

In[11]:=
```
RowReduce[%]
```

Out[11]//MatrixForm=

$$\begin{pmatrix} 1 & 0 & 0 & 1 \\ 0 & 1 & 0 & \frac{3}{2} \\ 0 & 0 & 1 & \frac{7}{2} \end{pmatrix}$$

This reduced form of the augmented matrix tells us that x1 = 1, x2 = 3/2 and x3 = 7/2.

Using **LinearSolve** :

The command **LinearSolve** provides a quick means for solving systems that have a single solution.

In[12]:=
```
LinearSolve[m, b]
```

Out[12]//MatrixForm=

$$\begin{pmatrix} 1 \\ \frac{3}{2} \\ \frac{7}{2} \end{pmatrix}$$

⇒ x1 = 1, x2 = 3/2 and x3 = 7/2.

Using **Determinant** and **Inverse** :

We can solve the system **m.x = b** for **x** by multiplying both sides on the left by **m^-1**, to get **x = m^-1.b**.

In[13]:=
```
Det[m]
```
Out[13]= -6

Since det ≠ 0 , therefore the system of linear equations has a unique solution.

In[14]:=
```
Inverse[m]
```
Out[14]//MatrixForm=

$$\begin{pmatrix} \frac{1}{3} & \frac{1}{3} & 0 \\ \frac{1}{2} & \frac{1}{2} & -\frac{1}{2} \\ \frac{1}{6} & -\frac{5}{6} & \frac{1}{2} \end{pmatrix}$$

In[15]:=
```
Inverse[m].b
```
Out[15]//MatrixForm=

$$\begin{pmatrix} 1 \\ \frac{3}{2} \\ \frac{7}{2} \end{pmatrix}$$

⇒ x1 = 1, x2 = 3/2 and x3 = 7/2.

Using **Solve** :

The **Solve** command takes a list of equations as its first argument and a list of variables as its second argument.

In[16]:=
```
Solve[m.x == b]
```
Out[16]//MatrixForm=

$$\begin{pmatrix} x1 \to 1 & x2 \to \frac{3}{2} & x3 \to \frac{7}{2} \end{pmatrix}$$

⇒ x1 = 1, x2 = 3/2 and x3 = 7/2.

An *inconsistent system* of equations has no solutions. If we use the **Solve** command on such a system, the output will be an empty set of curly brackets:

$$z_1 + z_2 + z_3 = 1$$
$$z_1 + z_2 + z_3 = 2$$
$$z_1 - z_2 - z_3 = -1$$

In[17]:=
```
m1 = {{1, 1, 1}, {1, 1, 1}, {1, -1, -1}}
```
Out[17]//MatrixForm=

$$\begin{pmatrix} 1 & 1 & 1 \\ 1 & 1 & 1 \\ 1 & -1 & -1 \end{pmatrix}$$

In[25]:=
```
z = {{z1}, {z2}, {z3}}
```

Out[25]//MatrixForm=

$$\begin{pmatrix} z1 \\ z2 \\ z3 \end{pmatrix}$$

In[20]:=
```
c = {{1}, {2}, {-1}}
```

Out[20]//MatrixForm=

$$\begin{pmatrix} 1 \\ 2 \\ -1 \end{pmatrix}$$

In[39]:=
```
Solve[m1.z == c]
```

Out[39]= {}

Therefore, the system has no solution.

If we use the **LinearSolve** command with an inconsistent system, a warning will be displayed that the Linear equation has no solution.

In[42]:=
```
LinearSolve[m1, c]
```

LinearSolve : Linear equation encountered that has no solution .

Out[42]=
```
LinearSolve[{{1, 1, 1}, {1, 1, 1}, {1, -1, -1}}, {{1}, {2}, {-1}}]
```

Therefore, the system has no solution.

However, if we **RowReduce**, we can see the inconsistency in the system.

In[40]:=
```
RowReduce[ArrayFlatten[{{m1, c}}]]
```

Out[40]//MatrixForm=

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

Therefore, the system has no solution as $0 \neq 1$.

The remaining possibility for a system of equations is that there are an *infinite number of solutions*.

$$2y_1 + 3y_2 - 4y_3 = 8$$
$$4y_1 + 6y_2 - 8y_3 = 16$$
$$y_1 - y_2 - y_3 = 1$$

In[32]:=
```
m2 = {{2, 3, -4}, {4, 6, -8}, {1, -1, -1}}
```

Out[32]//MatrixForm=

$$\begin{pmatrix} 2 & 3 & -4 \\ 4 & 6 & -8 \\ 1 & -1 & -1 \end{pmatrix}$$

In[33]:=
```
y = {{y1}, {y2}, {y3}}
```

Out[33]//MatrixForm=

$$\begin{pmatrix} y1 \\ y2 \\ y3 \end{pmatrix}$$

In[34]:=
```
d = {{8}, {16}, {1}}
```

Out[34]//MatrixForm=

$$\begin{pmatrix} 8 \\ 16 \\ 1 \end{pmatrix}$$

Be very careful when using the **LinearSolve** command. In a system having an infinite number of solutions it will return only one of them, giving no warning that there are others. In this example it returns only the solution where x3 = 0:

In[35]:=
```
LinearSolve[m2, d]
```

Out[35]//MatrixForm=

$$\begin{pmatrix} \frac{11}{5} \\ \frac{6}{5} \\ 0 \end{pmatrix}$$

⇒ y1 = 11/5, y2 = 6/5 and y3 = 0.

The **Solve** command nicely displays the solution set in this situation.

In[38]:=
```
Solve[m2.y == d]
```

Out[38]//MatrixForm=

$$\left(y2 \to \frac{4}{7} + \frac{2\,y1}{7} \quad y3 \to -\frac{11}{7} + \frac{5\,y1}{7} \right)$$

⇒ y2 = (4+2y1)/7 and y3 = (-11+5y1)/7.

RowReduce gives the solution with little possibility for confusion.

In[41]:=
```
RowReduce[ArrayFlatten[{{m2, d}}]]
```

Out[41]//MatrixForm=

$$\begin{pmatrix} 1 & 0 & -\frac{7}{5} & \frac{11}{5} \\ 0 & 1 & -\frac{2}{5} & \frac{6}{5} \\ 0 & 0 & 0 & 0 \end{pmatrix}$$

⇒ y1 = (11+7y3)/5 and y2 = (6+2y3)/5.

Therefore, we should be very careful using the command **LinearSolve** unless we know that we have a nonsingular matrix and hence a single solution. To check this, we can use the **Det** command, keeping in mind that a singular matrix has determinant zero. When in doubt it is best to use **RowReduce** to find the solution of the equations.

Homogenous systems of Linear Equations

A system of equations of the form **mx=0**, where **m** is the coefficient matrix, **x** is a column vector of variables, and **0** is the zero vector, is called homogeneous. Note that **x=0** is a solution to any homogeneous system. Now suppose **m** is a square matrix. Such a system of linear equations has a *unique solution* if and only if **m** is nonsingular. Hence, we see that if **m** is nonsingular, a homogeneous system will have only the *trivial* solution **x=0**, while if **m** is singular the system will have an *infinite* number of solutions. The set of all solutions to a homogeneous system is called the *null space* of **m**:

In[43]:=
```
m3 = {{0, 2, 2, 4}, {1, 0, -1, -3}, {2, 3, 1, 1}, {-2, 1, 3, -2}}
```

Out[43]//MatrixForm=

$$\begin{pmatrix} 0 & 2 & 2 & 4 \\ 1 & 0 & -1 & -3 \\ 2 & 3 & 1 & 1 \\ -2 & 1 & 3 & -2 \end{pmatrix}$$

In[44]:=
```
w = {{w1}, {w2}, {w3}, {w4}}
```

Out[44]//MatrixForm=

$$\begin{pmatrix} w1 \\ w2 \\ w3 \\ w4 \end{pmatrix}$$

In[45]:=
```
e = {{0}, {0}, {0}, {0}}
```

Out[45]//MatrixForm=

$$\begin{pmatrix} 0 \\ 0 \\ 0 \\ 0 \end{pmatrix}$$

In[46]:=
```
Det[m3]
```

Out[46]= 0

In[47]:=
```
RowReduce[ArrayFlatten[{{m3, e}}]]
```

Out[47]//MatrixForm=

$$\begin{pmatrix} 1 & 0 & -1 & 0 & 0 \\ 0 & 1 & 1 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 0 \end{pmatrix}$$

This reduced form of the augmented matrix tells us that x1 = x3, x2 = -x3, and x4 = 0.